Lesson 33: Chebyshev and Padé

```
> restart;
```

Chebyshev approximations in Maple

Maple has commands to produce Chebyshev series approximations: **chebpade** and **chebyshev** in the **numapprox** package.
We can use **chebpade** to produce an approximation to a given degree N, or chebyshev to attempt an approximation with error at most ε.

```
> with(numapprox):
f0:= x -> arctan(2*x+1);
```

$$f0 := x \rightarrow \arctan(2x+1)$$

```
> ChebApp1:= chebyshev(f0(x),x=-1..1,10^(-8));
```

$$\begin{aligned} ChebApp1 := {} & 0.452278447151191\, T(0,x) + 1.05817102727149\, T(1,x) \\ & - 0.272019649514069\, T(2,x) - 0.0288114945860669\, T(3,x) \\ & + 0.0581710272714922\, T(4,x) - 0.0179445424708701\, T(5,x) \\ & - 0.00573045751510579\, T(6,x) + 0.00696037216927648\, T(7,x) \\ & - 0.00164411429807495\, T(8,x) - 0.00112182104832474\, T(9,x) \\ & + 0.000974197112900738\, T(10,x) - 0.000137741903096520\, T(11,x) \\ & - 0.000215989116121847\, T(12,x) + 0.000142833053852007\, T(13,x) \\ & - 0.0000041307919421178 9\, T(14,x) - 0.0000408504603294581\, T(15,x) \\ & + 0.0000209405094970062\, T(16,x) + 0.00000233904648048977\, T(17,x) \\ & - 0.00000757640505946219\, T(18,x) + 0.00000296610644775531\, T(19,x) \\ & + 9.10669348976225\, 10^{-7}\, T(20,x) - 0.00000137447961971678\, T(21,x) \\ & + 3.88083734932629\, 10^{-7}\, T(22,x) + 2.43382924141505\, 10^{-7}\, T(23,x) \\ & - 2.43009422324076\, 10^{-7}\, T(24,x) + 4.24644368876864\, 10^{-8}\, T(25,x) \\ & + 5.66240774303582\, 10^{-8}\, T(26,x) - 4.16389150932986\, 10^{-8}\, T(27,x) \\ & + 2.45005887196057\, 10^{-9}\, T(28,x) + 1.21978131469271\, 10^{-8}\, T(29,x) \\ & - 6.85347846873698\, 10^{-9}\, T(30,x) + 2.49134438606034\, 10^{-9}\, T(32,x) \end{aligned}$$

The degree here will be 32.

```
> ChebApp2:= unapply(eval(%, T = orthopoly[T]),x);
```

$$\begin{aligned} ChebApp2 := x \rightarrow {} & 1.000000037081167\, x - 1.0000000153842041\, x^2 \\ & + 0.6666591956157818\, x^3 + 0.0000045105985637672 32\, x^4 - 0.7995573186639089\, x^5 \\ & + 1.333023991736615\, x^6 - 1.1550260071032648\, x^7 + 0.009486846158102136\, x^8 \\ & + 1.9662715620203932\, x^9 - 3.3627545726113643\, x^{10} + 1.0840520481328828\, x^{11} \\ & + 1.749441661514954\, x^{12} + 6.788026279520803\, x^{13} - 3.419954018503308\, x^{14} \\ & - 59.793397931109084\, x^{15} + 62.407875437755564\, x^{16} + 168.2943788434238\, x^{17} \\ & - 245.38418408181653\, x^{18} - 274.1524290199788\, x^{19} + 519.5925542990054\, x^{20} \end{aligned}$$

$$+ 292.0367553821713\, x^{21} - 712.5174102539618\, x^{22} - 208.385163612622\, x^{23} + 668.6857133975562\, x^{24} + 96.72552739008381\, x^{25} - 430.5951982022936\, x^{26} - 26.53320041376544\, x^{27} + 183.07810919181904\, x^{28} + 3.274325534298167\, x^{29} - 46.48040388069266\, x^{30} + 5.350121330601184\, x^{32} + 0.7853981633782545$$

```
> plot(f0(x)-ChebApp2(x),x=-1..1);
```



Computing Chebyshev polynomials

The Chebyshev polynomials $T(n, x)$ can be defined by $T(n, \cos(t)) = \cos(n\, t)$. But how should we calculate them for a particular x?

It turns out that these polynomials satisfy a recurrence relation very much reminiscent of the Fibonacci numbers:

$T(n, x) = 2\, x\, T(n-1, x) - T(n-2, x)$, with initial values $T(0, x) = 1$, $T(1, x) = x$.

```
> cos(n*t) = 2*cos(t)*cos((n-1)*t) - cos((n-2)*t);
```

$$\cos(n\, t) = 2 \cos(t) \cos((n-1)\, t) - \cos((n-2)\, t) \qquad \textbf{(2.1)}$$

```
> expand(%);
```

$$\cos(n\, t) = \cos(n\, t) \qquad \textbf{(2.2)}$$

This could be used with a **for** loop to calculate as many values of $T(n, x)$ as you need.

```
> V[0]:= 1; V[1]:= .234;
  for count from 2 to 6 do
    V[count]:= 2*.234*V[count-1] - V[count-2]
  end do;
```

$$V_0 := 1$$
$$V_1 := 0.234$$
$$V_2 := -0.890488$$
$$V_3 := -0.650748384$$
$$V_4 := 0.5859377563$$
$$V_5 := 0.9249672539$$
$$V_6 := -0.1530530815 \qquad (2.3)$$

Oh, but is it numerically stable?

```
> rsolve(T(n)=2*x*T(n-1)-T(n-2),T(n));
```

$$\frac{1}{2} \frac{1}{\sqrt{x^2-1}\left(-x+\sqrt{x^2-1}\right)}\left(\left(-T(0)-T(1)\,x+T(1)\sqrt{x^2-1}+2\,x^2\,T(0)-2\,x\,T(0)\sqrt{x^2-1}\right)\left(-\frac{1}{-x+\sqrt{x^2-1}}\right)^n\right) + \frac{1}{2} \frac{1}{\sqrt{x^2-1}\left(-x-\sqrt{x^2-1}\right)}\left(\left(T(0)+T(1)\,x+T(1)\sqrt{x^2-1}-2\,x^2\,T(0)-2\,x\,T(0)\sqrt{x^2-1}\right)\left(-\frac{1}{-x-\sqrt{x^2-1}}\right)^n\right)$$

The important things here are the $\left(-\frac{1}{-x-\sqrt{x^2-1}}\right)^n$ and $\left(-\frac{1}{-x+\sqrt{x^2-1}}\right)^n$: the solution is a linear combination of these, say $a\left(-\frac{1}{-x-\sqrt{x^2-1}}\right)^n + b\left(-\frac{1}{-x+\sqrt{x^2-1}}\right)^n$, with coefficients a and b that depend on x but not on n. A small error in, say, $T(m, x)$ would produce an error of this form in $T(n, x)$ for $n > m$. If that grew as $n \to \infty$ it would be bad news. Fortunately, for x in the interval [-1,1] (which is where we want it), it doesn't grow. $x^2 - 1 \le 0$, so $-x-\sqrt{x^2-1}$ and $-x+\sqrt{x^2-1}$ are complex numbers, and they have absolute value 1:
$\left|-x + i\sqrt{1-x^2}\right| = \sqrt{x^2+1-x^2} = 1.$
So the error doesn't grow much as $n \to \infty$. Let's try an example, with $x = 0.8$.

```
> y[0]:= 1; y[1]:= 0.8;
  for count from 2 to 100 do
    y[count]:= 2*0.8*y[count-1] - y[count-2]
  end do:
```

$$y_0 := 1$$

$$y_1 := 0.8$$

Here are the errors.

```
> seq([n,y[n]-cos(n*arccos(0.8))],n=1..100);
```

$[1, 0.], [2, 4.\ 10^{-10}], [3, -4.\ 10^{-10}], [4, -1.\ 10^{-10}], [5, 0.], [6, -2.\ 10^{-10}], [7, -4.\ 10^{-10}], [8, 3.\ 10^{-10}], [9, 1.\ 10^{-10}], [10, 2.\ 10^{-10}], [11, 1.\ 10^{-10}], [12, -3.\ 10^{-10}], [13, -1.4\ 10^{-9}], [14, -1.1\ 10^{-9}], [15, -2.\ 10^{-10}], [16, 1.2\ 10^{-9}], [17, 2.3\ 10^{-10}], [18, -9.\ 10^{-10}], [19, -9.\ 10^{-10}], [20, 5.\ 10^{-10}], [21, -5.3\ 10^{-9}], [22, -5.15\ 10^{-9}], [23, -3.0\ 10^{-9}], [24, -4.\ 10^{-10}], [25, 5.\ 10^{-10}], [26, -1.\ 10^{-10}], [27, -1.40\ 10^{-9}], [28, -2.0\ 10^{-9}], [29, -1.2\ 10^{-9}], [30, -2.9\ 10^{-9}], [31, -4.1\ 10^{-9}], [32, -3.1\ 10^{-9}], [33, -1.0\ 10^{-9}], [34, 6.\ 10^{-10}], [35, 3.\ 10^{-10}], [36, -1.9\ 10^{-9}], [37, -3.8\ 10^{-9}], [38, 2.4\ 10^{-9}], [39, -8.\ 10^{-10}], [40, -1.9\ 10^{-9}], [41, -9.\ 10^{-10}], [42, 1.0\ 10^{-9}], [43, 2.0\ 10^{-9}], [44, 7.\ 10^{-10}], [45, -2.7\ 10^{-9}], [46, 3.5\ 10^{-9}], [47, 1.6\ 10^{-9}], [48, -4.\ 10^{-10}], [49, -6.\ 10^{-10}], [50, 8.\ 10^{-10}], [51, 2.9\ 10^{-9}], [52, 4.2\ 10^{-9}], [53, 3.1\ 10^{-9}], [54, -2.\ 10^{-10}], [55, 2.5\ 10^{-9}], [56, 9.7\ 10^{-10}], [57, -1.1\ 10^{-9}], [58, -2.0\ 10^{-9}], [59, -7.\ 10^{-10}], [60, 2.1\ 10^{-9}], [61, 5.01\ 10^{-9}], [62, 5.9\ 10^{-9}], [63, 7.\ 10^{-10}], [64, 2.2\ 10^{-9}], [65, 1.5\ 10^{-9}], [66, -6.7\ 10^{-10}], [67, -2.5\ 10^{-9}], [68, -2.7\ 10^{-9}], [69, 0.], [70, 4.2\ 10^{-9}], [71, -2.6\ 10^{-9}], [72, 2.\ 10^{-10}], [73, 1.7\ 10^{-9}], [74, 1.0\ 10^{-9}], [75, -1.1\ 10^{-9}], [76, -3.4\ 10^{-9}], [77, -4.0\ 10^{-9}], [78, -2.1\ 10^{-9}], [79, 2.3\ 10^{-9}], [80, -2.3\ 10^{-9}], [81, -1.\ 10^{-10}], [82, 1.8\ 10^{-9}], [83, 2.0\ 10^{-9}], [84, -2.\ 10^{-10}], [85, -3.6\ 10^{-9}], [86, -5.9\ 10^{-9}], [87, -5.3\ 10^{-9}], [88, -2.2\ 10^{-9}], [89, -2.5\ 10^{-9}], [90, -1.1\ 10^{-9}], [91, 1.0\ 10^{-9}], [92, 2.1\ 10^{-9}], [93, 1.0\ 10^{-9}], [94, -2.3\ 10^{-9}], [95, -5.6\ 10^{-9}], [96, 1.8\ 10^{-9}], [97, -8.\ 10^{-10}], [98, -2.1\ 10^{-9}], [99, -1.5\ 10^{-9}], [100, 7.8\ 10^{-10}]$

Approximation using Chebyshev polynomials

So how could we have used the Chebyshev approximation for our function with this method of evaluating the polynomials?

It would be more efficient to get the coefficients in a table.

```
> for nn from 0 to 32 do
     CC[nn]:= coeff(ChebApp1,T(nn,x))
  end do:
```

Now to make a procedure that takes x, evaluates the $T(i, x)$ and evaluates **ChebApp2**.
The results will be accumulated in a variable R.
Each time we produce one of the $T(i, x)$ we add the appropriate contribution to R.

```
> ChebAppProc:= proc(x::numeric)
  local i, R, T;
  R:= CC[0] + x*CC[1];
  T[0]:= 1; T[1]:= x;
  for i from 2 to 32 do
```

```
    T[i]:= 2*x*T[i-1]-T[i-2];
    R:= R + T[i]*CC[i];
  end do;
  R
  end;
```

$ChebAppProc := \mathbf{proc}(x::numeric)$

local i, R, T;

$R := CC[0] + x * CC[1]$;

$T[0] := 1$;

$T[1] := x$;

for i **from** 2 **to** 32 **do**

$T[i] := 2 * x * T[i-1] - T[i-2]$; $R := R + T[i] * CC[i]$

end do;

R

end proc

```
> ChebAppProc(0.8); f0(0.8);
```

1.20362249250701

1.203622493 **(3.1)**

This will be more useful than the previous way of doing it only if we make sure to call **ChebAppProc** with numeric arguments, rather than the symbolic variable x. That's why I put in the type specification **x::numeric**.

For example, this would be wrong:

```
> plot(f0(x)-ChebAppProc(x),x=-1..1);
Error, invalid input: ChebAppProc expects its 1st argument, x,
to be of type numeric, but received x
```

What we can do instead is plot a function (or the difference of two functions) rather than an expression.

```
> plot(f0 - ChebAppProc, -1 .. 1);
```



Padé Approximation

Instead of using a polynomial approximation, it's sometimes better to use a rational function, i.e. the quotient of two polynomials. Consider a power series $f(x) = \sum_{k=0}^{\infty} c_k x^k$. A Padé approximant of type (m, n) is a rational function $\frac{p(x)}{q(x)}$ where p and q are polynomials of degrees at most m and n respectively, $q(0) \neq 0$, and the Maclaurin series of $\frac{p(x)}{q(x)}$ agrees with $f(x)$ for as many terms as possible, say the coefficient of x^j in $f(x) - \frac{p(x)}{q(x)}$ is 0 for $j \leq r$.

Multiplying by $q(x)$, this says the coefficient of x^j in $f(x)\,q(x) - p(x)$ is 0 for $j \leq r$. That would give us $r+1$ linear equations in $m+n+2$ unknowns, the coefficients of $q(x)$ and $p(x)$.

But multiplying both p and q by a nonzero constant won't change $\frac{p(x)}{q(x)}$, so we can assume that $q(0) = 1$. Now we just have $m+n+1$ unknowns. So we might hope that we can take $r = m+n$. Usually (but not always) this works. Sometimes it must be smaller.

```
> ft:= convert(taylor(f0(x),x,11),polynom);
```

$$ft := \frac{1}{4}\pi + x - x^2 + \frac{2}{3}x^3 - \frac{4}{5}x^5 + \frac{4}{3}x^6 - \frac{8}{7}x^7 + \frac{16}{9}x^9 - \frac{16}{5}x^{10} \quad (4.1)$$

```
> p5:= add(a[j]*x^j, j=0..5);
```

$$p5 := a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 \quad (4.2)$$

```
> q5:= 1 + add(b[j]*x^j, j=1..5);
```

$$q5 := 1 + b_1 x + b_2 x^2 + b_3 x^3 + b_4 x^4 + b_5 x^5 \quad (4.3)$$

```
> eqs:= [seq(coeff(ft*q5-p5,x,j)=0, j=0..10)];
```

$$eqs := \left[\frac{1}{4}\pi - a_0 = 0, \frac{1}{4}\pi b_1 + 1 - a_1 = 0, \frac{1}{4}\pi b_2 + b_1 - 1 - a_2 = 0, \frac{1}{4}\pi b_3 + \frac{2}{3} + b_2 - b_1 - a_3 = 0, -b_2 + \frac{2}{3}b_1 + \frac{1}{4}\pi b_4 + b_3 - a_4 = 0, b_4 - \frac{4}{5} + \frac{2}{3}b_2 + \frac{1}{4}\pi b_5 - b_3 - a_5 = 0, \frac{2}{3}b_3 - b_4 - \frac{4}{5}b_1 + b_5 + \frac{4}{3} = 0, \frac{2}{3}b_4 - \frac{8}{7} - b_5 - \frac{4}{5}b_2 + \frac{4}{3}b_1 = 0, \frac{4}{3}b_2 - \frac{8}{7}b_1 - \frac{4}{5}b_3 + \frac{2}{3}b_5 = 0, \frac{16}{9} - \frac{8}{7}b_2 - \frac{4}{5}b_4 + \frac{4}{3}b_3 = 0, -\frac{16}{5} + \frac{4}{3}b_4 - \frac{8}{7}b_3 - \frac{4}{5}b_5 + \frac{16}{9}b_1 = 0\right] \quad (4.4)$$

```
> solve(eqs);
```

$$\left\{a_0 = \frac{1}{4}\pi, a_1 = \frac{5}{4}\pi + 1, a_2 = \frac{25}{9}\pi + 4, a_3 = \frac{61}{9} + \frac{10}{3}\pi, a_4 = \frac{50}{9} + \frac{15}{7}\pi, a_5 = \frac{1744}{945} + \frac{37}{63}\pi, b_1 = 5, b_2 = \frac{100}{9}, b_3 = \frac{40}{3}, b_4 = \frac{60}{7}, b_5 = \frac{148}{63}\right\} \quad (4.5)$$

```
> f55:= eval(p5/q5,%);
```

$$f55 := \left(\frac{1}{4}\pi + \left(\frac{5}{4}\pi + 1\right)x + \left(\frac{25}{9}\pi + 4\right)x^2 + \left(\frac{61}{9} + \frac{10}{3}\pi\right)x^3 + \left(\frac{50}{9} + \frac{15}{7}\pi\right)x^4 + \left(\frac{1744}{945} + \frac{37}{63}\pi\right)x^5\right) \Big/ \left(1 + 5x + \frac{100}{9}x^2 + \frac{40}{3}x^3 + \frac{60}{7}x^4 + \frac{148}{63}x^5\right) \quad (4.6)$$

Similarly, for a (4,5) approximant:

```
> p4:= add(a[j]*x^j, j=0..4);
```

$$p4 := a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 \quad (4.7)$$

```
> eqs:= [seq(coeff(ft*q5-p4,x,j)=0, j=0..9)];
```

$$eqs := \left[\frac{1}{4}\pi - a_0 = 0, \frac{1}{4}\pi b_1 + 1 - a_1 = 0, \frac{1}{4}\pi b_2 + b_1 - 1 - a_2 = 0, \frac{1}{4}\pi b_3 + \frac{2}{3} + b_2 - b_1 - a_3 = 0, -b_2 + \frac{2}{3}b_1 + \frac{1}{4}\pi b_4 + b_3 - a_4 = 0, b_4 - \frac{4}{5} + \frac{2}{3}b_2 + \frac{1}{4}\pi b_5 - b_3 = 0, \frac{2}{3}b_3 - b_4 - \frac{4}{5}b_1 + b_5 + \frac{4}{3} = 0, \frac{2}{3}b_4 - \frac{8}{7} - b_5 - \frac{4}{5}b_2 + \frac{4}{3}b_1 = 0, \frac{4}{3}b_2 - \frac{8}{7}b_1 - \frac{4}{5}b_3 + \frac{2}{3}b_5 = 0, \frac{16}{9} - \frac{8}{7}b_2 - \frac{4}{5}b_4 + \frac{4}{3}b_3 = 0\right] \quad (4.8)$$

```
> solve(eqs);
```

$$\left\{a_0=\frac{1}{4}\pi,\ a_1=\frac{1}{9}\,\frac{1823\,\pi+435\,\pi^2+1440}{160+51\,\pi},\ a_2=\frac{1}{9}\,\frac{3537\,\pi+720\,\pi^2+4016}{160+51\,\pi},\ a_3=\frac{2}{63}\,\frac{15848+11397\,\pi+2025\,\pi^2}{160+51\,\pi},\ a_4=\frac{1}{189}\,\frac{38944+24960\,\pi+4005\,\pi^2}{160+51\,\pi},\ b_1=\frac{4}{9}\,\frac{1364+435\,\pi}{160+51\,\pi},\ b_2=\frac{64}{3}\,\frac{47+15\,\pi}{160+51\,\pi},\ b_3=\frac{8}{7}\,\frac{704+225\,\pi}{160+51\,\pi},\ b_4=\frac{20}{63}\,\frac{832+267\,\pi}{160+51\,\pi},\ b_5=-\frac{64}{105\,(160+51\,\pi)}\right\} \tag{4.9}$$

```
> f45:= eval(p4/q5,%);
```

$$f45 := \left(\frac{1}{4}\pi+\frac{1}{9}\,\frac{(1823\,\pi+435\,\pi^2+1440)\,x}{160+51\,\pi}+\frac{1}{9}\,\frac{(3537\,\pi+720\,\pi^2+4016)\,x^2}{160+51\,\pi}+\frac{2}{63}\,\frac{(15848+11397\,\pi+2025\,\pi^2)\,x^3}{160+51\,\pi}+\frac{1}{189}\,\frac{(38944+24960\,\pi+4005\,\pi^2)\,x^4}{160+51\,\pi}\right)\Bigg/\left(1+\frac{4}{9}\,\frac{(1364+435\,\pi)\,x}{160+51\,\pi}+\frac{64}{3}\,\frac{(47+15\,\pi)\,x^2}{160+51\,\pi}+\frac{8}{7}\,\frac{(704+225\,\pi)\,x^3}{160+51\,\pi}+\frac{20}{63}\,\frac{(832+267\,\pi)\,x^4}{160+51\,\pi}-\frac{64}{105}\,\frac{x^5}{160+51\,\pi}\right) \tag{4.10}$$

```
> plot([f45-f0(x),f55-f0(x)],x=-1..1,-0.01..0.01,colour=[red,
green]);
```



The (5,5) approximant actually has a vertical asymptote at $x = -1$.

```
> factor(f55);
```

$$\frac{1}{60}\left(945\,\pi + 4725\,x\,\pi + 3780\,x + 10500\,x^2\,\pi + 15120\,x^2 + 25620\,x^3 + 12600\,x^3\,\pi + 21000\,x^4 + 8100\,x^4\,\pi + 6976\,x^5 + 2220\,x^5\,\pi\right) \Big/ \left((x+1)\left(148\,x^4 + 392\,x^3 + 448\,x^2 + 252\,x + 63\right)\right) \tag{4.11}$$

```
> fsolve(denom(f45));
```

$$873.2447298 \tag{4.12}$$

```
> plot([f45-f0(x),f55-f0(x)], x=-1..10, -0.01 .. 0.01, colour=
[red,blue]);
```



The Padé approximant might approximate the function well over a much larger interval than the Maclaurin series approximation. Often some experimentation is needed to tell what are the best *m* and *n* to use.

Maple computes Padé approximants with the **pade** command in the **numapprox** package.

```
> f45 - pade(f0(x),x=0,[4,5]); normal(%);
```

$$\left(\frac{1}{4}\pi + \frac{1}{9}\,\frac{\left(1823\,\pi + 435\,\pi^2 + 1440\right)x}{160 + 51\,\pi} + \frac{1}{9}\,\frac{\left(3537\,\pi + 720\,\pi^2 + 4016\right)x^2}{160 + 51\,\pi} + \frac{2}{63}\,\frac{\left(15848 + 11397\,\pi + 2025\,\pi^2\right)x^3}{160 + 51\,\pi} + \frac{1}{189}\,\frac{\left(38944 + 24960\,\pi + 4005\,\pi^2\right)x^4}{160 + 51\,\pi}\right) \Bigg/ \left(1 + \frac{4}{9}\,\frac{\left(1364 + 435\,\pi\right)x}{160 + 51\,\pi} + \frac{64}{3}\,\frac{\left(47 + 15\,\pi\right)x^2}{160 + 51\,\pi} + \frac{8}{7}\,\frac{\left(704 + 225\,\pi\right)x^3}{160 + 51\,\pi} + \frac{20}{63}\,\frac{\left(832 + 267\,\pi\right)x^4}{160 + 51\,\pi} - \frac{64}{105}\,\frac{x^5}{160 + 51\,\pi}\right) - \Bigg(105\,\Bigg(640\,\pi + 204\,\pi^2 + \Bigg(2560 + \frac{29168}{9}\,\pi$$

$$+\frac{2320}{3}\pi^2\Big)x+\Big(\frac{64256}{9}+6288\pi+1280\pi^2\Big)x^2+\Big(\frac{72448}{9}+\frac{121568}{21}\pi+\frac{7200}{7}\pi^2\Big)x^3+\Big(\frac{623104}{189}+\frac{133120}{63}\pi+\frac{7120}{21}\pi^2\Big)x^4\Big)\Big)\Big/\Big(268800+85680\pi+\Big(324800\pi+\frac{3055360}{3}\Big)x+(537600\pi+1684480)x^2+(432000\pi+1351680)x^3+\Big(142400\pi+\frac{1331200}{3}\Big)x^4-1024x^5\Big)$$

$$0$$

The Padé approximant is usually even faster to compute than the corresponding polynomial, if done using a continued fraction form.

```
> hornerform(evalf(ft));
```

$$0.7853981635+(1+(-1.+(0.6666666667+(-0.8000000000+(1.333333333+(-1.142857143+(1.777777778-3.200000000\,x)\,x^2)\,x)\,x)\,x^2)\,x)\,x)\,x$$

```
> codegen[cost](%);
```

$$8\ additions+10\ multiplications$$

```
> evalf(f45);
```

$$(0.7853981635+3.976556903\,x+7.714780569\,x^2+7.102102055\,x^3+2.592221119\,x^4)\Big/(1.+3.789869957\,x+6.270590943\,x^2+5.035298762\,x^3+1.656399762\,x^4-0.001903445996\,x^5) \quad \textbf{(4.13)}$$

```
> codegen[cost](evalf(f45));
```

$$9\ additions+25\ multiplications+divisions \quad \textbf{(4.14)}$$

```
> convert(evalf(f45),confrac,x);
```

$$-1361.856929\Bigg/\Bigg(x-872.9508369-256.6468394\Bigg/\Bigg(x+0.02064064193+\frac{0.3786762167}{x+1.490883027-\frac{0.03455720024}{x-1.486021057+\frac{4.778170673}{x+2.714272171}}}\Bigg)\Bigg)$$

```
> codegen[cost](%);
```

$$5\ divisions+9\ additions$$

This gives me an excuse to say something about continued fractions.

Continued fractions

A **continued fraction** is an expression of the form

```
> b[0]+a[1]/(b[1]+a[2]/(b[2]+a[3]/(b[3]+a[4]/`...`)));
```

$$b_0+\cfrac{a_1}{b_1+\cfrac{a_2}{b_2+\cfrac{a_3}{b_3+\cfrac{a_4}{\ldots}}}} \tag{5.1}$$

where the ... might continue forever (an infinite continued fraction) or might stop at some point (a terminating continued fraction). It is a **simple continued fraction** if all $a_n = 1$. If you delete everything with a subscript $> n$, you get a finite expression which is called the nth convergent of the continued fraction. Thus the zeroth, first and second convergents are

```
> [b[0], b[0] + a[1]/b[1], b[0]+a[1]/(b[1]+a[2]/(b[2]))];
```

$$\left[b_0, b_0+\frac{a_1}{b_1}, b_0+\cfrac{a_1}{b_1+\cfrac{a_2}{b_2}}\right] \tag{5.2}$$

Of course, these could be simplified to ordinary quotients:

```
> normal(%);
```

$$\left[b_0, \frac{b_0\, b_1+a_1}{b_1}, \frac{b_0\, b_1\, b_2+b_0\, a_2+a_1\, b_2}{b_1\, b_2+a_2}\right] \tag{5.3}$$

The **numtheory** package contains several commands for dealing with continued fractions.

```
> with(numtheory):
```

The main one is **cfrac**. For a general continued fraction:

```
> cfrac([b[0],[a[1],b[1]],[a[2],b[2]]]);
```

$$\frac{b_0\, b_1\, b_2+b_0\, a_2+a_1\, b_2}{b_1\, b_2+a_2} \tag{5.4}$$

Or for a simple continued fraction:

```
> cfrac([b[0],b[1],b[2],b[3],b[4]]) = normal(b[0]+1/(b[1]+1/(b
  [2]+1/(b[3]+1/b[4]))));
```

$$\frac{b_0\, b_1\, b_2\, b_3\, b_4+b_0\, b_1\, b_2+b_0\, b_1\, b_4+b_0\, b_3\, b_4+b_0+b_2\, b_3\, b_4+b_2+b_4}{b_1\, b_2\, b_3\, b_4+b_1\, b_2+b_1\, b_4+b_3\, b_4+1} = \frac{b_0\, b_1\, b_2\, b_3\, b_4+b_0\, b_1\, b_2+b_0\, b_1\, b_4+b_0\, b_3\, b_4+b_0+b_2\, b_3\, b_4+b_2+b_4}{b_1\, b_2\, b_3\, b_4+b_1\, b_2+b_1\, b_4+b_3\, b_4+1} \tag{5.5}$$

The convergents may also be obtained with the **nthconver** command.

```
> seq(nthconver([b[0],[a[1],b[1]],[a[2],b[2]],[a[3],b[3]]],j),
  j=0..3);
```

$$b_0, \frac{b_0\, b_1+a_1}{b_1}, \frac{b_0\, b_1\, b_2+a_1\, b_2+a_2\, b_0}{b_1\, b_2+a_2}, \frac{b_0\, b_1\, b_2\, b_3+a_1\, b_2\, b_3+a_2\, b_0\, b_3+a_3\, b_0\, b_1+a_1\, a_3}{b_1\, b_2\, b_3+a_2\, b_3+a_3\, b_1} \tag{5.6}$$

Or for a simple continued fraction:

```
> seq(nthconver([b[0],b[1],b[2],b[3],b[4]],j),j=0..4);
```

$$b_0, \frac{b_0 b_1 + 1}{b_1}, \frac{b_0 b_1 b_2 + b_0 + b_2}{b_1 b_2 + 1}, \frac{b_0 b_1 b_2 b_3 + b_0 b_1 + b_0 b_3 + b_2 b_3 + 1}{b_1 b_2 b_3 + b_1 + b_3}, \frac{b_0 b_1 b_2 b_3 b_4 + b_0 b_1 b_2 + b_0 b_1 b_4 + b_0 b_3 b_4 + b_2 b_3 b_4 + b_0 + b_2 + b_4}{b_1 b_2 b_3 b_4 + b_1 b_2 + b_1 b_4 + b_3 b_4 + 1} \quad (5.7)$$

To see the continued fraction typeset as a continued fraction:

```
> CFRAC([b[0],[a[1],b[1]],[a[2],b[2]],[a[3],b[3]]]);
```

$$b_0 + \cfrac{a_1}{b_1 + \cfrac{a_2}{b_2 + \cfrac{a_3}{b_3}}} \quad (5.8)$$

The numerators and denominators satisfy a recurrence relation. If the nth convergent is $\frac{P_n}{Q_n}$, then

- $P_n = b_n P_{n-1} + a_n P_{n-2}$ with $P_0 = b_0$, $P_{-1} = 1$
- $Q_n = b_n Q_{n-1} + a_n Q_{n-2}$ with $Q_0 = 1$, $Q_{-1} = 0$

```
> p[0]:= b[0]; p[-1]:= 1; q[0]:= 1; q[-1]:= 0;
  for nn from 1 to 5 do
    p[nn]:= expand(b[nn]*p[nn-1] + a[nn]*p[nn-2]);
    q[nn]:= expand(b[nn]*q[nn-1] + a[nn]*q[nn-2]);
  end do;
```

$$p_0 := b_0$$
$$p_{-1} := 1$$
$$q_0 := 1$$
$$q_{-1} := 0$$
$$p_1 := b_1 b_0 + a_1$$
$$q_1 := b_1$$
$$p_2 := b_2 b_1 b_0 + b_2 a_1 + a_2 b_0$$
$$q_2 := b_2 b_1 + a_2$$
$$p_3 := b_3 b_2 b_1 b_0 + b_3 b_2 a_1 + b_3 a_2 b_0 + a_3 b_1 b_0 + a_3 a_1$$
$$q_3 := b_3 b_2 b_1 + b_3 a_2 + a_3 b_1$$
$$p_4 := b_4 b_3 b_2 b_1 b_0 + b_4 b_3 b_2 a_1 + b_4 b_3 a_2 b_0 + b_4 a_3 b_1 b_0 + b_4 a_3 a_1 + a_4 b_2 b_1 b_0 + a_4 b_2 a_1 + a_4 a_2 b_0$$
$$q_4 := b_4 b_3 b_2 b_1 + b_4 b_3 a_2 + b_4 a_3 b_1 + a_4 b_2 b_1 + a_4 a_2$$
$$p_5 := b_5 b_4 b_3 b_2 b_1 b_0 + b_5 b_4 b_3 b_2 a_1 + b_5 b_4 b_3 a_2 b_0 + b_5 b_4 a_3 b_1 b_0 + b_5 b_4 a_3 a_1$$

$+ b_5 a_4 b_2 b_1 b_0 + b_5 a_4 b_2 a_1 + b_5 a_4 a_2 b_0 + a_5 b_3 b_2 b_1 b_0 + a_5 b_3 b_2 a_1 + a_5 b_3 a_2 b_0 + a_5 a_3 b_1 b_0 + a_5 a_3 a_1$

$$q_5 := b_5 b_4 b_3 b_2 b_1 + b_5 b_4 b_3 a_2 + b_5 b_4 a_3 b_1 + b_5 a_4 b_2 b_1 + b_5 a_4 a_2 + a_5 b_3 b_2 b_1 + a_5 b_3 a_2 + a_5 a_3 b_1 \quad (5.9)$$

```
> nthconver([b[0],[a[1],b[1]],[a[2],b[2]],[a[3],b[3]]],3) - p[3]/q[3];
```

$$0 \quad (5.10)$$

Continued fractions for numbers

Consider any positive number x. We can represent it by a simple continued fraction with integer elements as follows. Let $b_0 = \text{floor}(x)$, so $0 \leq x - b_0 < 1$. If that is 0, then $x = b_0$. Otherwise $x = b_0 + \frac{1}{x_1}$ with $x_1 > 1$. Let $b_1 = \text{floor}(x_1) \geq 1$. Again, if $b_1 = x_1$, then $x = b_0 + \frac{1}{b_1}$, otherwise $x = b_0 + \cfrac{1}{b_1 + \cfrac{1}{x_2}}$ etc.

The continued fraction for x terminates if and only if x is a rational number. For example:

```
> cfrac([3,7,110,3]); CFRAC([3,7,110,3]);
```

$$\frac{7291}{2320}$$

$$3 + \cfrac{1}{7 + \cfrac{1}{110 + \cfrac{1}{3}}} \quad (6.1)$$

```
> cfrac(7291/2320);
```

$$3 + \cfrac{1}{7 + \cfrac{1}{110 + \cfrac{1}{3}}} \quad (6.2)$$

```
> x1:= 1/(7+1/(110+1/3));
```

$$x1 := \frac{331}{2320} \quad (6.3)$$

```
> 7291/2320 -3 ;
```

$$\frac{331}{2320} \quad (6.4)$$

```
> floor(2320/331);
```

$$7 \quad (6.5)$$

```
> 2320/331 - 7;
```

$$\frac{3}{331} \quad (6.6)$$

```
> 331/3 - 110;
```

$$\frac{1}{3} \tag{6.7}$$

Maple commands introduced in this lesson:

pade in **numapprox** package
hornerform in **numapprox** package
cost in **codegen** package
convert(..., **confrac)**
numtheory package
cfrac in **numtheory** package
nthconver in **numtheory** package
CFRAC